

# Funkcje wyszukiwania i adresu – INDEKS

**Mariusz Jankowski**

autor strony internetowej poświęconej Excelowi i programowaniu w VBA;

**Bogdan Gilarski**

właściciel firmy szkoleniowej Perfect And Practical;

Pytania: czytelnicy.controlling@infor.pl

W tym opracowaniu kontynuujemy problematykę funkcji wyszukiwania i adresu. W numerze 3/2010 miesięcznika opisaliśmy zastosowanie PODAJ.POZYCJĘ. Tym razem zajmiemy się funkcją INDEKS.

**P**odstawowym zadaniem, które realizuje funkcja INDEKS, jest zwrócenie pojedynczej wartości (lub tablicy) z podanego zbioru danych. Zbiorem może być w tym przypadku zakres komórek, wpisana ręcznie tablica lub kilka niezależnych od siebie obszarów z danymi. INDEKS potrafi w wyniku zwrócić także adres, ale tę funkcjonalność omówimy w następnym opracowaniu.

Funkcja ta doskonale współpracuje z inną funkcją wyszukiwania i adresu – PODAJ.POZYCJĘ dlatego, że w pewnym sensie spełnia odwrotną rolę. PODAJ.POZYCJĘ, jak napisaliśmy w poprzednim artykule (nr 3/2010), określa pozycję (kolejność) wystąpienia konkretnej wartości na jednowymiarowej tablicy danych; INDEKS natomiast zwraca w wyniku wartość na podstawie jej pozycji (robi to za pomocą dwóch argumentów *Nr\_wiersza* i *Nr\_kolumny*) – w przeciwieństwie do PODAJ.POZYCJĘ może operować na dwuwymiarowej tabeli danych.

## Budowa funkcji – składnia tablicowa

Funkcja INDEKS posiada dwie składnie – tablicową (składającą się z trzech argumentów) i odwołańową (zawierającą cztery parametry). Różnice dotyczą przede wszystkim pierwszego argumentu funkcji – w wersji tablicowej możemy zaznaczyć (wpisać) tylko pojedynczą tablicę danych, w wersji odwołańowej natomiast kilka niepowiązanych ze sobą zakresów komórek. Składnia odwołańowa zawiera także dodatkowy argument, który określa numer zakresu, do któ-

rego będziemy się odwoływać w formule. Zarówno jedna jak i druga wersja może zwracać pojedyncze wartości, fragmenty tablic, jak i całe tablice danych.

Prześledźmy budowę funkcji w wersji tablicowej (z jedną tabelą danych). *Tablica* to po prostu tabela (zakres komórek w arkuszu) lub tablica (wpisana ręcznie), z której chcemy pobrać dane. *Nr\_wiersza* to numer wiersza tablicy. *Nr\_kolumny* to argument opcjonalny (w sytuacji gdy tabela zawiera więcej niż jedną kolumnę danych, należy podać konkretny numer kolumny, z której chcemy pobrać dane).

Jak widzimy na **widoku 1**:

- funkcja działa prawidłowo zarówno dla tablicy jednowymiarowej (formuła 1, 2, 4 i 5) jak i dla tablicy dwuwymiarowej (formuła 3),
- dla jednowymiarowej tablicy pionowej (formuła 1) wymagany jest drugi argument funkcji (*Nr\_wiersza*), natomiast trzeci argument (*Nr\_kolumny*) jest opcjonalny (tablica zawiera tylko jedną kolumnę – można wpisać także 0 lub 1),
- dla jednowymiarowej tablicy poziomej (formuła 2) wymagany jest trzeci argument funkcji (*Nr\_kolumny*), natomiast drugi argument (*Nr\_wiersza*) jest opcjonalny (tablica zawiera tylko jeden wiersz – można wpisać także 0 lub 1),
- funkcja działa prawidłowo dla tabeli dwuwymiarowej, jeżeli poprawnie podane są argumenty: *Nr\_wiersza* i *Nr\_kolumny* (formuła 3),
- funkcja generuje błąd #ADR! jeżeli argumenty *Nr\_wiersza* lub *Nr\_kolumny* przyjmują wartości wyższe niż odpowiednio: liczba wierszy

w tabeli (formuła 4), liczba kolumn w tabeli (formuła 5).

Funkcja INDEKS pozwala nie tylko na pobieranie pojedynczych wartości, ale części lub nawet całej tabeli źródłowej. Chcąc pobrać cały wiersz, musimy ustawić wartość drugiego argumentu na 0, chcąc pobrać całą kolumnę – analogicznie musimy ustawić wartość trzeciego argumentu na 0. Równoczesne ustawienie drugiego i trzeciego argumentu funkcji na 0 pozwala zwrócić w wyniku całą tabelę źródłową. Ponieważ w wyniku zwracana jest tablica wartości, formułę należy zatwierdzić w sposób tablicowy (Ctrl+Shift+Enter).

**Widok 2** zawiera dwie formuły tablicowe:

- pierwsza formuła (komórka D3) odwołuje się do pierwszego wiersza z danymi; ustawienie zera dla trzeciego argumentu powoduje, że w domyśle mają być zwrócone wszystkie kolumny z danymi; formuła w wyniku zwraca zatem dwie wartości (tablica pozioma),
- druga formuła (komórka D6) w domyślne zwraca całą tabelę, ale ponieważ umieściliśmy ją tylko w czterech komórkach, zwraca fragment tablicy.

## Budowa funkcji – składnia odwołań

W wersji odwołaniowej budowa funkcji wygląda nieco inaczej. Pierwszy argument *Odwołanie* to odwołanie do pojedynczej tabeli lub wielu tabel (lista zakresów musi być ujęta w nawiasy). Drugi i trzeci argument wyglądają tak samo jak w wersji tablicowej, czyli *Nr\_wiersza* i *Nr\_kolumny*. Składnia ta posiada także czwarty argument *Nr\_obszaru*, który pozwala na wybór jednego spośród wybranych w pierwszym argumentcie funkcji – obszarów.

Należy jednak zaznaczyć, że funkcja w tej formie posiada pewne ograniczenia – obszary podawane

w pierwszym argumentcie (nawet jeżeli są to nazwane obszary) muszą być zlokalizowane w tym samym arkuszu, w którym znajduje się funkcja. Z tego też powodu ta wersja zastosowania funkcji INDEKS() jest mało popularna i raczej niepolecana w przypadku bardziej zaawansowanych projektów.

**Widok 3** zawiera cztery formuły (aby ułatwić ich zrozumienie, wprowadziliśmy nazwy: *Europa* – odnosi się do zakresu A3:B6, *Azja* – odnosi się do zakresu D3:D5, *Afryka* – odnosi się do zakresu A8:D8):

- pierwsza formuła odwołuje się do nazwy *Europa* – pobiera wartość z czwartego wiersza, drugiej kolumny zakresu – czyli *Hiszpanię*,
- druga formuła odwołuje się do wszystkich trzech zakresów zawierających nazwy państw. Ostatni argument ustawiony na 2 wskazuje, że bierzemy pod uwagę tylko zakres *Azja* (D3:D5). Trzeci wiersz zakresu to *Indie*, zatem taka wartość zostaje zwrócona w wyniku. Argument *Nr\_kolumny* jest ustawiony na 1, ponieważ tablica składa się tylko z jednej kolumny, jednak pominięcie te-

**Widok 1**

### Wersja tablicowa funkcji INDEKS – pojedyncze wartości

	A	B	C	D	E	F	G
1	Wersja tablicowa funkcji INDEKS - pojedyncze wartości						
2							
3	Rok	Sprzedaż [mln.]					
4	2005	8.70		2008	=INDEKS(\$A\$4:\$A\$8;4)		
5	2006	5.88		8.7	=INDEKS(\$A\$4:\$B\$4;;2)		
6	2007	2.24		5.88	=INDEKS(\$A\$4:\$B\$8;2;2)		
7	2008	4.01		#ADR!	=INDEKS(\$A\$4:\$A\$8;6;1)		
8	2009	6.72		#ADR!	=INDEKS(\$A\$4:\$A\$8;1;3)		
9							

**Widok 2**

### Wersja tablicowa funkcji INDEKS – tablica wartości

	A	B	C	D	E	F
1	Wersja tablicowa funkcji INDEKS - tablica wartości					
2						
3	Rok	Sprzedaż [mln.]		2005	8.7	
4	2005	8.70		=INDEKS(\$A\$4:\$B\$8;1;0)		
5	2006	5.88				
6	2007	2.24		2005	8.7	
7	2008	4.01		2006	5.88	
8	2009	6.72		=INDEKS(\$A\$4:\$B\$8;0;0)		
9						

go argumentu lub wpisanie zera daje dokładnie taki sam efekt (pobierany jest wtedy cały wiersz),

- trzecia formuła zwraca cały trzeci wiersz zakresu *Europa* – formuła musi zostać zatwierdzona w sposób tablicowy,
- czwarta formuła działa podobnie. Tym razem odwołujemy się do zakresu *Afryka* – pobieramy cały pierwszy (jedyne) wiersz. Ponieważ na wynik zarezerwowaliśmy tylko dwie komórki, w wyniku otrzymujemy połowę tablicy.

### Zastosowanie funkcji – wyszukiwanie w tablicy pionowej

Funkcja INDEKS jest na tyle elastyczną funkcją, że pozwala na pobieranie danych nawet

Widok 3

Wersja odwołaniowa funkcji INDEKS						
	A	B	C	D	E	F
1	Wersja odwołaniowa funkcji INDEKS					
2						
3	Polska	Czechy		Japonia		
4	Rosja	Francja		Chiny		
5	Niemcy	Włochy		Indie		
6	Grecja	Hiszpania				
7						
8	Egipt	RPA	Nigeria	Kamerun		
9						
10	Hiszpania		=INDEKS(Europa;4;2;1)			
11	Indie		=INDEKS((Europa;Azja;Afryka);3;1;2)			
12	Niemcy	Włochy	=INDEKS((Europa;Azja;Afryka);3;0;1)			
13	Egipt	RPA	=INDEKS((Europa;Azja;Afryka);1;0;3)			
14						

Widok 4

Wyszukiwanie w tablicy pionowej							
	A	B	C	D	E	F	G
1	Wyszukiwanie w tablicy pionowej						
2							
3	Kwartał 1	styczeń	luty	marzec			
4	Kwartał 2	kwiecień	maj	czerwiec			
5	Kwartał 3	lipiec	sierpień	wrzesień			
6	Kwartał 4	październik	listopad	grudzień			
7							
8	maj	Kwartał 2					
9		=INDEKS(kwartaly;MAX(JEŻELI(miesiące=\$A8;WIERSZ(miesiące);0))-2)					
10							
11	grudzień	Kwartał 4					
12		=INDEKS(kwartaly;MAX(JEŻELI(miesiące=\$A11;WIERSZ(miesiące);0))-2)					
13							

z takich tabel, których układ nie jest do końca prawidłowy. Widok 4 zawiera listę kwartałów (wprowadzonych do pierwszej kolumny arkusza) i listę miesięcy (wpisanych aż do trzech różnych kolumn) – zadanie polega na przyporządkowaniu kwartału z kolumny A, konkretnemu miesiącowi.

Aby ułatwić zrozumienie zadania, wprowadziliśmy dwie nazwy:

*Kwartaly* – odnosi się do zakresu A3:A6,

*Miesiące* – odnosi się do zakresu B3:D6.

Jak widać, formuła tablicowa (Ctrl+Shift+Enter) działa prawidłowo zarówno dla maja jak i dla grudnia. Przystudujmy formułę dla maja:

=INDEKS(kwartaly;MAX(JEŻELI(miesiące=\$A8;WIERSZ(miesiące);0))-2)

- $JEŻELI(miesiące=\$A8;WIERSZ(miesiące);0)$  – najpierw porównujemy dane z zakresu *Miesiące* ze słowem „maj”. Każdy miesiąc znajduje się na liście tylko raz, pewne jest zatem, że formuła zwróci wartość PRAWDA dokładnie w jednym przypadku. Chcemy, aby funkcja JEŻELI zwracała w wyniku numer wiersza komórki (dla PRAWDA) lub 0 (dla FAŁSZ). Funkcja tworzy tablicę wartości składającą się z zer i jednej czwórki (numer wiersza komórki z tekstem „maj”).

- $MAX(JEŻELI(miesiące=\$A8;WIERSZ(miesiące);0))-2$  – funkcja MAX pobiera największą wartość z tablicy – czyli 4. Ponieważ zakres komórek *Miesiące* zaczyna się od trzeciego wiersza, musimy odjąć 2 – chcemy, aby w wyniku podany został numer wiersza zakresu, w którym znajduje się komórka z tekstem „maj”. W wyniku otrzymujemy 2.

- Następnie za pomocą funkcji INDEKS pobieramy

drugi wiersz zakresu *Kwartały* (tablica jednokolumnowa). Wynikiem jest tekst *Kwartał 2*.

## Zastosowanie funkcji – wyszukiwanie w tablicy poziomej

Widok 5 przedstawia podobny przypadek, z tą różnicą, że tym razem chcemy pobrać dane z tablicy poziomej. Tak jak poprzednio, wprowadziliśmy dwie nazwy, które pozwalają na łatwiejsze zrozumienie sposobu działania formuł tablicowych:

*Quarters* – odnosi się do zakresu A3:D3,

*Months* – odnosi się do zakresu A4:D6.

Jak widać, formuła tablicowa (Ctrl+Shift+Enter) działa prawidłowo zarówno dla lutego jak i dla lipca. Przetudujmy formułę dla lipca:

$INDEKS(quarters;;MAX(JEŻELI(months=\$A11;NR.KOLUMNY(months);0)))$

- $JEŻELI(months=\$A11;NR.KOLUMNY(months);0)$  – najpierw porównujemy dane z zakresu *Months* ze słowem „lipiec”. Każdy miesiąc znajduje się na liście tylko raz, tak więc podobnie jak w poprzednim przykładzie, formuła zwróci wartość PRAWDA dokładnie w jednym przypadku. Chcemy, aby funkcja JEŻELI zwracała w wyniku numer kolumny komórki (dla PRAWDA) lub 0 (dla FAŁSZ). W rezultacie funkcja tworzy tablicę wartości składającą się z zer i jednej trójki.

- $MAX(JEŻELI(months=\$A11;NR.KOLUMNY(months);0))$  – funkcja MAX pobiera największą wartość z tablicy – czyli 3.

- Następnie za pomocą funkcji INDEKS pobieramy trzecią kolumnę zakresu *Quarters* (tablica jednowierszowa). Wynikiem jest tekst *Kwartał 3*.

- W formule został pominięty drugi argument, czyli *Nr\_wiersza*, dlatego, że zakres *Quarters* to czteroelementowa tablica pozioma. Ustawienie tego argumentu na 0 (cały wiersz) lub na 1 (pierwszy wiersz) daje identyczne rezultaty.

Widok 5

Wyszukiwanie w tablicy poziomej							
	A	B	C	D	E	F	G
1	Wyszukiwanie w tablicy poziomej						
2							
3	Kwartał 1	Kwartał 2	Kwartał 3	Kwartał 4			
4	styczeń	kwiecień	lipiec	październik			
5	luty	maj	sierpień	listopad			
6	marzec	czerwiec	wrzesień	grudzień			
7							
8	luty	Kwartał 1					
9		=INDEKS(quarters;;MAX(JEŻELI(months=\\$A8;NR.KOLUMNY(months);0)))					
10							
11	lipiec	Kwartał 3					
12		=INDEKS(quarters;;MAX(JEŻELI(months=\\$A11;NR.KOLUMNY(months);0)))					
13							

Widok 6

Pobranie danych znajdujących się po lewej stronie							
	A	B	C	D	E	F	G
1	Pobranie danych znajdujących się po lewej stronie						
2							
3	Rok	Sprzedaż [mln.]		Rok	Sprzedaż		
4	2005	8,70		2008	4,01		
5	2006	5,88		=INDEKS(rok;PODAJ.POZYCJE(\\$E4;sprzedaż;0))			
6	2007	2,24					
7	2008	4,01					
8	2009	6,72					
9							

## Współpraca z funkcją PODAJ.POZYCJĘ

Zagnieżdżenie funkcji PODAJ.POZYCJĘ wewnątrz INDEKS pozwala na tworzenie bardzo szybkich, elastycznych i użytecznych formuł zwracających potrzebne wartości z tabeli źródłowej. Kombinacja tych funkcji pozwala pobierać dane z tabel zorganizowanych w różny sposób. Dlatego też często z powodzeniem bywa wykorzystywana jako alternatywa wobec funkcji WYSZUKAJ.PIONOWO. INDEKS + PODAJ.POZYCJĘ pozwala m.in. na: wyszukiwanie wartości na podstawie kilku różnych kryteriów, wyszukiwanie krzyżowe (na podstawie kryteriów znajdujących się w pierwszym wierszu i pierwszej kolumnie), a także na pobranie wartości znajdującej się w kolumnie po lewej stronie względem szukanej. I temu ostatniemu przypadkowi chcielibyśmy się przyjrzeć dokładniej, ponieważ jest to jeden z najczęściej spotykanych problemów do rozwiązania.

## Pobranie danych znajdujących się po lewej stronie

Funkcja WYSZUKAJ.PIONOWO umożliwia pobranie danych znajdujących się po prawej stronie od kolumny, w której znajduje się szukana wartość. Nie ma natomiast możliwości, aby pobrać wartość, która znajduje się po lewej stronie. W takiej sytuacji z pomocą przychodzi kombinacja funkcji INDEKS + PODAJ.POZYCJĘ.

W powyższym przykładzie chcemy pobrać wartość z kolumny A (Rok), na podstawie wartości z kolumny B (Sprzedaż). Znając wysokość sprzedaży (4.01), chcemy pobrać rok. Rozwiązaniem jest formuła:

`=INDEKS(rok;PODAJ.POZYCJĘ($E4;sprzedaż;0))`

- Najpierw pobierana jest pozycja liczby 4.01. Ostatni argument ustawiony na 0 wskazuje, że szukamy dokładnego przypisania. 4.01 znajduje się na czwartej pozycji zakresu Sprzedaż (nagłówki nie są brane pod uwagę), zatem drugi argument funkcji INDEKS przyjmuje wartość 4.

- Wartością, która znajduje się w czwartym wierszu tabeli *Rok*, jest 2008, i rzeczywiście jest to prawidłowy wynik.

## Wnioski

Jak wspomnieliśmy na wstępie, zdecydowana większość zastosowań funkcji INDEKS polega na ściągnięciu pojedynczej wartości z obszaru (tabeli) poprzez wskazanie punktu przecięcia (za pomocą dwóch argumentów funkcji: *Nr\_wiersza*, i *Nr\_kolumny*). Funkcja posiada dwie składnie:

- tablicową, która pozwala pobrać pojedynczą wartość lub tablicę wartości, z określonego w pierwszym argumente funkcji zakresu,
- odwołaniową (bardzo rzadko stosowaną w praktyce), która umożliwia wybór obszaru, z którego ma zostać pobrana wartość, pod warunkiem że wszystkie zakresy znajdują się w tym samym arkuszu, co funkcja INDEKS.

Uniwersalność tej funkcji polega na tym, że doskonale sprawdza się w przypadkach pobierania informacji z tabel zorganizowanych w różny sposób. Trudno jest sobie wyobrazić szybkie i niezawodne przetwarzanie danych za pomocą formuł bez wykorzystywania funkcji INDEKS.



### Analiza danych biznesowych z wykorzystaniem MS Access

21 - 23.04.2010 Warszawa

Celem szkolenia jest dostarczenie wiedzy o tym, jak Access może wspomóc analizę danych biznesowych tam, gdzie Excel osiąga granice swoich możliwości. Uczestnicy poszerzą swoje wiadomości o sposobach dostępu i analizie danych zawartych w różnych formatach i bazach za pomocą najpopularniejszej platformy dostępu i tworzenia baz danych MS Access.

### Controlling - nowoczesne kierunki rozwoju

17 - 19.05.2010 Warszawa

W trakcie szkolenia zostaną przedstawione wybrane narzędzia z zakresu nowoczesnego controllingu. Ich zakres został dobrany w taki sposób, aby wskazać najnowsze tendencje rozwoju tej dziedziny zarządzania.

### Excel dla zaawansowanych. Platforma analiz biznesowych z elementami budżetowania.

24 - 27.05.2010 Warszawa

Kurs ma silne ukierunkowanie na zastosowania w biznesie, analizach finansowych i w zarządzaniu.

#### KONTAKT:

tel. (022) 530 42 59, faks (022) 530 43 09  
szkolenia@infor.pl, www.szkolenia.infor.pl